

Centre for Ecology and Hydrology  
Bush Estate  
Penicuik  
Midlothian  
EH26 0QB

Merlewood Research and Development Paper  
Number 37

A BASIC version of WORLD2

J. N. R. Jeffers, FIS, AMBIM

R & D 72/37



## Introduction

In his book on "World Dynamics", Forrester (1971) presents an explicit version of his dynamic model of the world. The model, which is based on theoretical concepts developed from the investigation of industrial and urban dynamics, interrelates population, capital investment, geographical space, natural resources, pollution, and food production. From these major sectors and their interactions appear to come the dynamics of change in the world system, at least in so far as "the world" can be reasonably regarded as an entity. Thus, rising population creates pressures to increase industrial output, grow more food, and occupy more land, and these factors, in turn, create larger populations. In time, growth of population, industrial output and agricultural output encounter limits as land and natural resources become exhausted, and the capacity of the earth to dissipate pollution becomes overloaded.

Forrester's model draws heavily upon two concepts of system structure. The most important of these concepts is that all effects take place within "feedback loops", i.e. closed paths that connect an "action" to an "effect" in such a way that future actions are influenced by past effects. The feedback may be positive or negative, the former generating increased growth in a system and the latter generating stability in seeking an equilibrium. But any feedback loop may itself be complex, so that a series of "actions" may be linked to a series of "effects", and vice versa, in a way which is difficult, if not impossible, to describe verbally. For this reason, systems models have to be constructed in mathematical terms so that the nature of the feedback and the relationships between variables can be made explicit.

The second important concept is that the relationships between the variables included in a system will seldom be linear, and the extent and nature of the non-linearity of these relationships has important effects on the behaviour of the model. Much of the overt simplification of mathematical models employed in many branches of science depends upon the assumption of linearity of relationships, and the escape from this constraint in the modelling process is one of the advantages sought from systems analysis.

WORLD2, from which Forrester drew many of his initial conclusions, is given explicitly in his book, so that it is possible to explore some alternative solutions to the problems he poses, and to test the importance of various basic assumptions directly on the same model. Various modifications of the model have taken place, even before it was published, but WORLD2 provides an interesting and reasonable starting point for a study of these developments, and for developments which other workers may wish to initiate. This paper presents a version of WORLD2 in the BASIC computer language and reports on the results of some simple experiments with the model as examples of the ease with which it can be manipulated on interactive computer systems.

### The BASIC model

Forrester presented his model in the notation used by DYNAMO, a computer program for translating mathematical models from a readily understood notation into tabulated and plotted results. The model consists of algebraic relationships that relate the variables one to another, and considerable ingenuity is usually required to translate these relationships into a form in which they can be handled sequentially by the computer. DYNAMO avoids many of the logical difficulties by providing direct facilities for defining levels, variables, and constants, and for defining functions by interpolation from relatively few fixed points.

The use of a program like DYNAMO is not without its disadvantages. First, it requires a large computer to operate satisfactorily, and much of the power and storage space of the computer is used to hold the translator and its ancillary functions, even when large parts of the program are not needed for a particular model. Second, the language places certain constraints on the formulation of the model, which is practically limited to a set of zero and first-order difference equations. Third, because of the size of the translator, it is usually necessary to run DYNAMO programs in a batch-processing mode rather than in an interactive, conversational mode. The facilities for experimentation may, therefore, be severely limited, especially if the programmer is remote from the computer.

Appendix I gives a version of Forrester's WORLD2 in the BASIC language. BASIC is a simple, conversational, computer language for scientific, business and educational applications, used to solve simple or complex mathematical problems and directed from a user's teletype terminal. Input and output to the program is through a teletype terminal, and the language can be readily used on time-sharing systems. Furthermore, the notation is at least as simple as that of DYNAMO, even if some of the refinements of the latter are less readily available. In compensation, there are few limitations on the form of the model which can be formulated.

Figure 1 shows the essential construction of the program for WORLD2, the numbers in the boxes giving the locations of the statements for each segment. (BASIC statements are always numbered to facilitate editing and alteration of the programs.) Essentially, after setting the initial conditions, the program consists of two series of iterations, allowing for a switch in various parameters of the model after 1970. Both series of iterations refer to the main subroutine, which contains the principal relationships postulated for WORLD2 and described by Forrester (1971). In this subroutine, the table look-up functions of DYNAMO are replaced by empirical curve-fitting through the points given by Forrester. The interpolated values will not, therefore, be exactly the same as those derived from DYNAMO functions, but they are sufficiently close for the model to behave in the same way, and the relationships are, in any case, defined somewhat arbitrarily.

Manipulation of the model, in its BASIC form, is achieved by changing the values of the coefficients, or by altering the form of the principal relationships in the main subroutine. Either of these changes can be made through the extensive editing facilities provided by the BASIC language, so that the consequences of the changes can be tested quickly and effectively on time-sharing computer systems. The form of the BASIC statements is necessarily as explicit as that for DYNAMO, but restrictions on the length of the names of variables in the BASIC language have made necessary some redefinition of the letter groups used to identify variables and constants in the model equations. The correspondence between Forrester's definitions and those used in the BASIC version of the model is summarised in Table 1.

Running the program of Appendix I in its unchanged form gives the basic behaviour of the world model as postulated by Forrester, in which industrialization and population are suppressed by falling natural resources. Table 2 gives the typical outputs from the model for levels of population, natural resources, capital investment, pollution ratio, and quality of life at 10-year intervals. As an example of the ease with which modification can be made to the basic program, the following alterations to the program result in Figure 2 which plots the same values diagrammatically:-

```

130
170 PRINT TAB (INT((P/1E8) + 0.5)); "P";
175 PRINT TAB (50 + INT ((R/1E10) + 0.5)); "R";
180 PRINT TAB (2 * INT ((C/1E9) + 0.5)); "C";
185 PRINT TAB (INT (Z1 + 0.5)); "Z";
187 PRINT TAB (INT ((40 * Q) + 0.5)); "Q"

330 PRINT TAB (INT ((P/1E8) + 0.5)); "P";
335 PRINT TAB (50 + INT ((R/1E10) + 0.5)); "R";
340 PRINT TAB (2 * INT ((C/1E9) + 0.5)); "C";
345 PRINT TAB (INT (Z1 + 0.5)); "Z";
347 PRINT TAB (INT ((40 * Q) + 0.5)); "Q"

```

Figure 2 corresponds to the similar figure given by Forrester, except that the scales for the individual variables have been deliberately omitted so as to focus attention on the general shape of the curves rather than on over-precise predictions for any one year.

Printing or plotting of other variables of the basic model may be achieved by similar modification of the BASIC statements 165-190 and 325-360. Note that, throughout this paper, the BASIC exponential form will be used to represent large numbers. Thus, 1.5E9 should be interpreted as  $1.5 \times 10^9$  or 1,500,000,000.

#### A test of some initial conditions

Partly as an illustration of the ease with which the BASIC model may be manipulated, a test of three of the initial conditions was carried out. One of the frequent criticisms of the concept of dynamic models is that they are not only dependent upon the correctness of the feedback relationships that are included in the model, but they are also extremely sensitive to small changes in the values assigned to the system levels. It is important to establish, therefore, the extent and nature of this sensitivity.

Three of the system levels were chosen for this preliminary test, namely, natural resources, capital investment and pollution, and the BASIC model was run with the combinations of initial parameters given in Table 3, and the resulting series of experiments represents a  $3^3$  factorial design. However, it quickly became apparent that the changes of the order planned in the initial level of pollution (Z) had no effect on the output from the model, and the experiment could therefore be reduced to the  $3^2$  factorial design for the combinations of the levels of natural resources (R) and capital investment (C).

The effects of the changes in these parameters on population (P), natural resources (NR), capital investment (CI), pollution ratio (POLR), and quality of life (QL) are shown diagrammatically in Figures 3, 4, 5, 6, and 7, respectively. No detailed analysis of these effects will be attempted in this paper, but it is clear that, while the general shape of the curves remains approximately similar, changes in the initial values of the parameters for natural resources and

capital investment have marked effects on the estimates, and particularly on the levels of natural resources and the pollution ratio. The model would appear to be relatively sensitive to joint changes in the initial values of some parameters, but not to others, and further exploration of this sensitivity is proceeding.

#### Effect of reductions in the usage ratio of natural resources

The basic behaviour of WORLD2, characterized in Table 2 and Figure 2, postulates that population and capital investment will grow until natural resources decline far enough to inhibit expansion. As resources decline still further, the world is unable to sustain the peak population. Population then declines along with capital investment and quality of life, which is dependent on material standard of living, food supply, crowding, and pollution, falls because of the pressures created by the shortage of natural resources.

Forrester (1971) suggested that, instead of allowing a limit to growth to be imposed by declining resources, technology might find ways to use the more plentiful materials, recycle the more scarce resources, and to increase sources of energy so that the depletion of resources is no longer the major constraint upon population growth. He demonstrated the effects of removing this constraint by changing the value of usage rate of natural resources (NRUN) after 1970. In other words, to see whether a more desirable future is created, we assume that technology maintains the standard of living by reducing the drain on expendable and irreplaceable resources.

Figure 8 gives the changes in population (P) associated with varying levels of the usage rate of natural resources after 1970, and the actual population values are summarized in Table 4. Reductions in usage rate to 0.90 have only a small effect on population levels, and a reduction of the usage rate to 0.85 also has a relatively small effect on population, except for the marked fall in population between 2050 and 2100 which was also associated with increases in the initial value of natural resources in the test of the sensitivity of the model to the input parameters. Reductions of the usage rate to 0.80 and below increase the peak value reached by the population, but also lead to increasingly rapid collapse of the population in subsequent years.

The mechanism behind this dramatic collapse of population levels as the constraint of limited resources is of some interest. With the reduced dependence on resources, pollution and capital investment rise until a pollution crisis is reached. Pollution then acts directly to reduce birth rate, increase death rate, and to depress food production. Table 5 gives the pollution ratios (POLR) corresponding to the usage rate of natural resources after 1970, and these ratios are shown diagrammatically in Figure 9. The rapid rise in the pollution ratios for usage rates less than 0.85 is particularly striking, and further illustrates the way in which, because of the relationships specified by this model, population and capital investment grow until they generate pollution at a rate beyond that which the environment can dissipate. The resulting pollution overloading results in a rapid decline in the level of the population and capital investment until the rate of pollution generation falls below the rate of pollution of absorption.

#### Progressive reduction of the usage of natural resources

Reduction in the usage rate of natural resources to a fixed proportion from an arbitrary date is clearly unrealistic. Even if ways could be found to reduce the demand per person made upon natural resources, for example, by recycling materials or by improved technology, the rate at which such a reduction could be made would at first be relatively small, and then, hopefully, increase

progressively. Tests were therefore carried out on the basic model in which the natural resource usage normal (NRUN) was set equal to other variables of the model after 1970.

In particular, four progressive reductions were tested:-

1. NRUN set equal to  $R/R_0$ , i.e. the usage normal set equal to the proportion of the current level of natural resources to the level of resources in 1900.
2. NRUN set equal to  $R/R_{1970}$ , i.e. the usage normal set equal to the proportion of the current level of natural resources to the level of resources in 1970.
3. NRUN set equal to QL, i.e. the usage normal set equal to the quality of life.
4. NRUN set equal to MSL, i.e. the usage normal set equal to the material standard of living.

The results of these modifications to the basic model are summarised in Tables 6 and 7, and in Figures 10 and 11. Only one of the modifications, where NRUN is set equal to MSL, shows any improvement over the basic model. The other three modifications all lead to higher peak populations followed by a rapid decline in population associated with a soaring pollution ratio. Setting NRUN equal to MSL (the material standard of living) after 1970 actually gives a slight improvement over the basic model by keeping the peak population marginally lower and reducing the rate of collapse, giving an estimated increase of  $0.3E9$  persons in 2100. The calculated values of population, natural resources, capital investment, pollution ratio, and quality of life for this "improved" model are given in Table 8, together with the calculated values of MSL. Thus, achievement of this modest improvement in the model would require reduction of usage normal NRUN to 0.99 by the year 2010, 0.76 by the year 2050, and 0.52 by the year 2090.

#### Changes in rates of substitution

Fisher and Pry (1971) have proposed a simple substitution model of technological change. The model is based on three assumptions:-

1. Many technological advances can be considered as competitive substitutions of one method of satisfying a need for another.
2. If a substitution has progressed as far as a few per cent, it will proceed to completion.
3. The fractional rate of fractional substitution of new for old is proportional to the remaining amount of the old left to be substituted.

They postulate that substitutions tend to proceed exponentially (i.e. with a constant percentage annual growth increment) in the early years, and to follow an S-shaped curve. The simplest such curve is characterized by two constants: the early growth rate and the time at which the substitution is half-complete.

The corresponding fraction substituted is given by:

$$f = \left(\frac{1}{2}\right) \sqrt{1 + \tanh \alpha (t - t_0)}$$

where  $\alpha$  is half the annual fractional growth in the early years and where  $t_0$  is the time at which  $f = 0.5$ .

Thus substitution model can be incorporated into the world dynamic model by adding the following statements to the BASIC program:

```
210 DEF FNA(A) = (EXP(A) - EXP(-A))/(EXP(A) + EXP(-A))
215 LET H = 0.026
220 LET T = 2070
311 LET X = (1/2) * (1 + FNA (H * (((I - 10) + J) - T)))
312 LET N5 = 1 - X
```

Thereafter, by changing the values of H and T in statements 215 and 220 respectively, it is possible to explore the consequences of changing these two parameters on the world model predictions, where H and T represent  $\alpha$  and  $t_0$ , respectively.

Beginning from values of H and T of 0.010 and 2100 respectively, which were derived by a rough approximation of the substitution rates of the improved model obtained by setting NRUN equal to MSL, an optimal gradient method (P. Wolfe, 1967) was used to search for the combination of values of these two parameters which gave the highest predicted population for the year 2100. A series of experiments was also carried out on the model subsequently to determine the shape of the response surface generated by changes in the values of the two parameters. Figure 12 gives the predicted populations in the year 2100 for critical values of H at each of several values of T, the predicted population falling off rapidly for all values of H when T is less than 2040.

In general, there is a particular range of values of H which gives the maximum values of predicted population in the year 2100 for any given value of T between 2110 and 2040. Furthermore, the maximum predicted population for any value of T increases as T increases until  $T = 2040$ , and then sharply declines as is illustrated in Figure 13. The value of H for which the population in the year 2100 is a maximum corresponding to each value of T is plotted in Figure 14.

In practical terms, allowing for the usage of natural resources to be slowed down by a sigmoid substitution model, as opposed to the arbitrary reduction illustrated by Forrester, results in a stabilization of the world population, after a sharp decline from a peak, at values considerably higher than those predicted for the year 2100 by the basic WORLD2 model. Provided that an optimum rate of annual fractional growth is chosen, the level at which the world population stabilizes is increased as the year in which the fraction of natural resource usage is 0.5 is brought earlier, reaching a maximum when  $T = 2040$ . A more rapid rate of substitution, however, results in a sharp decline in the predicted population for the year 2100. The calculated values of the optimum substitution rates for each value of T are given in Table 9 and the corresponding predicted populations are given in Table 10. These results are also shown diagrammatically in Figures 15 and 16, respectively.

The calculated values of population, natural resources, capital investment, pollution ratio, and quality of life for the optimum substitution strategy, for which  $H = 0.220$  and  $T = 2040$ , are given in Table 11 and Figure 17. The population rises to a peak of  $5.40E9$  in 2020 and then falls to a stable level of  $4.34E9$  - a reduction of  $0.56E9$  from the peak population. Natural resources decline rapidly, before being stabilized at  $4.35E9$  resource units, a little under half



the units available in 1900. Capital investment increases steadily to slightly more than 9E9 capital units in the year 2040, and thereafter declines only slowly. The pollution ratio climbs steadily to about 7 and then declines slowly, while the quality of life falls fairly rapidly from its peak in 1940 to 0.61 in 2040, and then rises slowly to 0.68 in 2100. Thus, while still not an ideal solution to the problem of world dynamics, the optimum substitution model nevertheless represents a considerable improvement over the basic model, as a result of only slight modification of the original relationships.

### Discussion

Forrester's WORLD2 model has been extensively criticised. Attention has been directed, in particular, to the sensitivity of the model to changes in some of the initial parameters, but much of the relevant criticism has also been directed to the lack of social feedback in the model. The model structure does not leave sufficient scope for man to intervene when the world system is seen to be developing in an undesirable direction. Oerlemans et al (1972) have suggested an extension of the WORLD2 model in which the total fraction of investment in agriculture and pollution control together is determined, and this fraction subsequently split into a "capital investment in agriculture" fraction and a "capital investment in pollution" fraction according to the ratio QLE/QLP. In this way, they were able to avoid the pollution crisis which Forrester suggests to be the consequence of reducing the natural resources usage normal (NRUN).

Similarly, Boyd (1972) added a new state variable, technology (T), and multipliers to express the effect of technology on the other state variables. The numerical values of the multipliers and constants introduced were intended to reflect, in a reasonable way, the technological optimist's faith that, for example, we will be able to find a substitute for any diminishing natural resource. In this way, he was able to produce results which are exactly what a technological optimist would predict. Technology increases productivity, which, in turn, increases the standard of living. This increase eventually drives birth rates low enough for a "Utopian" equilibrium to be reached.

It is, however, worth noting that both these modifications of the WORLD2 model required extensive additions to the already long list of parameters, variables, and levels to achieve a technologically optimistic solution. As this paper has shown, only a very simple model of substitution for non-renewable natural resources is required to achieve a considerable improvement in the predicted population in the year 2100, even if the optimum substitution rate does not completely achieve a technologically optimistic solution. It seems likely that similar exploration of the original model may find particular combinations of the parameters which, combined with optimum, but possibly different, substitution rates, will lead to a more optimistic solution.

Forrester's demonstration that reduced usage rate of natural resources leads to a pollution crisis was unrealistic. It is inconceivable that the usage rate could be reduced to 25 per cent of its original value in 1970 immediately, and any reduction in the usage rate would at first be small, increasing at first slowly, and then more rapidly, as the technological problems were solved, until it became progressively harder to find substitutes for the remaining fraction of natural resources. Not surprisingly, therefore, the model reacts to this violent extrapolation by becoming unstable. Nevertheless, there are some feasible rates of reduction of the usage rate of natural resources which do not lead to instability of the model, and some of these rates give markedly more optimistic estimates of population levels, without any need to multiply the assumptions of the model.

Exploration of the values of the two parameters introduced into the substitution model demonstrates very clearly that the rate at which recycling and substitution of non-renewable natural resources is introduced into the world system may have a decisive effect upon the dynamics of the system. For any kind of improved solution, the fraction substituted must not be greater than 0.2 by the year 2030, but, after that date, the more rapid the increase in the growth of the fraction substituted, the better the solution. The optimum solution, if none of the other parameters of the model is changed, is derived by making no appreciable substitution until the year 2030, achieving a 0.50 fractional substitution by 2040, and almost complete substitution by 2050. Such a solution fits very well with the concept of a "technological breakthrough" which can be rapidly implemented by industry, but it is important to note that any such "breakthrough" before 2030 (or at least implemented before 2030) could lead to sub-optimal solutions.

The important point to be made in this preliminary examination of WORLD2 is that mathematical models of any reasonable degree of complexity need to be explored carefully. Such models are frequently only a little less complex in their implications and behaviour than real systems, and it will usually be necessary to employ the same statistical devices that are used to explore real systems in order to understand the behaviour of model systems. In the work reported in this paper, for example, basic concepts of experimental design, response surface analysis, and evolutionary operation were used to gain an understanding of the response of the model to changes in only a few of its many parameters. Much further exploration remains to be done before it is possible to say whether or not WORLD2 can be regarded as a useful approximation to the real system. A few cursory experiments based on unrealistic manipulations are not a sufficient reason for advocating wholesale modifications to the basic assumptions, especially where these modifications have the effect of "multiplying the entities needlessly". It has been suggested that Ockham's well-known advice should not be elevated to the status of an inviolable methodological principle (Skellam, 1972), but it is perhaps worth emphasising that much of the purpose of having models is that they can be experimented with more easily than the real system.

However, the main purpose of this paper has been to demonstrate that models of the level of complexity of WORLD2 can as readily be implemented in a general-purpose language such as BASIC as in the highly-specialized modelling language of DYNAMO. Not only is the BASIC version capable of being run on simple conversational computer systems, so that experimentation with the model is quick and easy to perform, but modifications to the model can be made so easily that the experimenter is able to follow through his ideas with practically no limitations on the form of the model. But, even more important, the use of BASIC is more economical of both core storage and time on the computer, enabling all of the work to be done on relatively cheap computer systems. Indeed, the use of languages like DYNAMO for dynamic modelling is an example of the wasteful use of resources that WORLD2 is intended to illustrate and, hopefully, to remedy.

## References

- R. Boyd, 1972. World dynamics: a note. Science 177 (August), 516-9.
- J. C. Fisher and R. H. Pry, 1971. A simple substitution model of technological change. Technological Forecasting and Social Change 3 (1), 75-88.
- J. W. Forrester, 1971. World dynamics. Wright-Allen Press Inc., Cambridge, USA.
- T. W. Oerlemans et al, 1972. World dynamics: social feedback may give hope for the future. Nature 238 (5362) 251-5.
- J. G. Skellam, 1972. Some philosophical aspects of mathematical modelling in empirical science with special reference to ecology. (In J.N.R. Jeffers (editor) Mathematical models in ecology. Proc. 12th Symposium of the British Ecological Society. Blackwell Scientific Publications, Oxford, England.)
- P. Wolfe, 1967) Methods of non-linear programming. (In J. Abadie (editor) Non-linear programming. North-Holland Publishing Co., Amsterdam, Holland.)

Table 1. Definition of symbols used in program

Group	Definition		BASIC variable
BR	Birth rate	(people/year)	B
BRCM	Birth rate from crowding multiplier	(dimensionless)	B3
BRFM	Birth rate from food multiplier	(dimensionless)	B1
BRMM	Birth rate from material multiplier	(dimensionless)	B2
BRPM	Birth rate from pollution multiplier	(dimensionless)	B4
CFIFR	Capital fraction indicated by food ratio	(dimensionless)	A1
CI	Capital investment	(capital units)	C
CIAF	Capital investment in agriculture fraction	(dimensionless)	C2
CIAFN	Capital investment in agriculture fraction normal	(")	C3
CIRA	Capital investment ratio in agriculture	(capital units/person)	C5
CID	Capital investment discard	(capital units/year)	C7
CIG	Capital investment generation	(capital units/year)	C6
CIM	Capital investment multiplier	(dimensionless)	C8
CIQR	Capital investment from quality ratio	(dimensionless)	A2
CIR	Capital investment ratio	(capital units/year)	C1
CR	Crowding ratio	(dimensionless)	C4
DR	Death rate	(people/year)	D
DRCM	Death rate from crowding multiplier	(dimensionless)	D4
DRFM	Death rate from food multiplier	(dimensionless)	D3
DRMM	Death rate from material multiplier	(dimensionless)	D1
DRPM	Death rate from pollution multiplier	(dimensionless)	D2
ECIR	Effective capital investment ratio	(dimensionless)	E
FC	Food coefficient	(dimensionless)	F5
FCM	Food from crowding multiplier	(dimensionless)	F3
FN	Food normal	(food units/person/year)	F6
FPCI	Food potential from capital investment	(")	F2
FPM	Food from pollution multiplier	(dimensionless)	F4
FR	Food ratio	(dimensionless)	F1
LA	Land area	(square kilometers)	L
MSL	Material standard of living	(dimensionless)	M
NR	Natural resources	(natural resource units)	R
NREM	Natural resource extraction multiplier	(dimensionless)	N1
NRFR	Natural resource fraction remaining	(dimensionless)	N2
NRMM	Natural resource from material multiplier	(dimensionless)	N4
NRUN	Natural resource usage normal	(natural resource units/person/year)	N5
NRUR	Natural resource usage rate	(natural resource units/year)	N3
P	Population	(people)	P
PDN	Population density normal	(people/square kilometer)	L(0)
POL	Pollution	(pollution units)	Z
POLA	Pollution absorption	(pollution units/year)	Z3
POLAT	Pollution absorption time	(years)	Z5
POLCM	Pollution from capital multiplier	(dimensionless)	Z4
POLG	Pollution generation	(pollution units/year)	Z2
POLR	Pollution ratio	(dimensionless)	Z1
QL	Quality of life	(satisfaction units)	Q
QLC	Quality of life from crowding	(dimensionless)	Q2
QLF	Quality of life from food	(dimensionless)	Q3
QLM	Quality of life from material	(dimensionless)	Q1
QLP	Quality of life from pollution	(dimensionless)	Q4

Table 2. Calculated values of population, natural resources, capital investment, pollution ratio, and quality of life for basic model

Year	PE9	RE9	CE9	POLR	Q
1910	1.67	893.19	.63	.05	.86
1920	1.87	883.8	.93	.09	1.02
1930	2.17	870.87	1.34	.15	1.11
1940	2.56	853.38	1.88	.24	1.13
1950	3	830.38	2.57	.37	1.12
1960	3.48	801.13	3.4	.57	1.07
1970	3.97	765.34	4.37	.85	1
1980	4.43	723.39	5.41	1.24	.93
1990	4.85	676.37	6.46	1.76	.85
2000	5.16	626.05	7.43	2.44	.78
2010	5.35	574.53	8.24	3.28	.72
2020	5.4	523.98	8.82	4.23	.67
2030	5.32	476.23	9.14	5.19	.63
2040	5.15	432.56	9.18	5.96	.6
2050	4.92	393.65	8.98	6.26	.58
2060	4.71	359.65	8.59	5.84	.57
2070	4.53	330.34	8.05	4.8	.54
2080	4.36	305.3	7.43	3.67	.51
2090	4.17	284.05	6.76	2.79	.49
2100	3.97	266.09	6.07	2.14	.46

Table 3. Selected combinations of initial parameters

Experiment Number	Natural resources RE9	Capital investment CE9	Pollution ZE9
1			0.1
2		0.3	0.2
3			0.3
4			0.1
5	800	0.4	0.2
6			0.3
7			0.1
8		0.5	0.2
9			0.3
10			0.1
11		0.3	0.2
12			0.3
13			0.1
14 *	900	0.4	0.2
15			0.3
16			0.1
17		0.5	0.2
18			0.3
19			0.1
20		0.3	0.2
21			0.3
22			0.1
23	1000	0.4	0.2
24			0.3
25			0.1
26		0.5	0.2
27			0.3

\* Represents the basic model parameter, i.e. the control.

Table 4. Changes in levels of population associated with reductions in usage rate of natural resources

Year	Population (PE 9) corresponding to usage rate (NRUN) of:							
	0.25	0.50	0.75	0.80	0.85	0.90	0.95	1.00
1910	1.67	1.67	1.67	1.67	1.67	1.67	1.67	1.67
1920	1.87	1.87	1.87	1.87	1.87	1.87	1.87	1.87
1930	2.17	2.17	2.17	2.17	2.17	2.17	2.17	2.17
1940	2.56	2.56	2.56	2.56	2.56	2.56	2.56	2.56
1950	3.00	3.00	3.00	3.00	3.00	3.00	3.00	3.00
1960	3.48	3.48	3.48	3.48	3.48	3.48	3.48	3.48
1970	3.97	3.97	3.97	3.97	3.97	3.97	3.97	3.97
1980	4.44	4.44	4.44	4.44	4.44	4.44	4.44	4.43
1990	4.90	4.88	4.86	4.86	4.86	4.85	4.85	4.85
2000	5.30	5.26	5.21	5.20	5.19	5.18	5.17	5.16
2010	5.63	5.55	5.46	5.44	5.42	5.40	5.37	5.35
2020	5.81	5.70	5.56	5.53	5.50	5.47	5.44	5.40
2030	5.65	5.62	5.51	5.47	5.44	5.40	5.36	5.32
2040	4.52	5.09	5.26	5.25	5.24	5.21	5.18	5.15
2050		3.49	4.77	4.86	4.92	4.94	4.94	4.92
2060			3.96	4.32	4.54	4.66	4.70	4.71
2070			2.69	3.63	4.18	4.44	4.52	4.53
2080			1.27	2.81	3.95	4.33	4.39	4.36
2090				1.98	3.94	4.26	4.24	4.17
2100				1.47	4.02	4.13	4.06	3.97

Table 5. Changes in POLR associated with values of NRUN

Year	Pollution ratio (POLR) corresponding to usage rate (NRUN) of:							
	0.25	0.50	0.75	0.80	0.85	0.90	0.95	1.00
1910	0.05	0.05	0.05	0.05	0.05	0.05	0.05	0.05
1920	0.09	0.09	0.09	0.09	0.09	0.09	0.09	0.09
1930	0.15	0.15	0.15	0.15	0.15	0.15	0.15	0.15
1940	0.24	0.24	0.24	0.24	0.24	0.24	0.24	0.24
1950	0.37	0.37	0.37	0.37	0.37	0.37	0.37	0.37
1960	0.57	0.57	0.57	0.57	0.57	0.57	0.57	0.57
1970	0.85	0.85	0.85	0.85	0.85	0.85	0.85	0.85
1980	1.24	1.24	1.24	1.24	1.24	1.24	1.24	1.24
1990	1.83	1.81	1.79	1.78	1.78	1.77	1.77	1.76
2000	2.70	2.61	2.53	2.51	2.50	2.48	2.46	2.44
2010	4.09	3.81	3.54	3.49	3.43	3.38	3.33	3.28
2020	6.54	5.69	4.91	4.76	4.62	4.49	4.35	4.23
2030	11.57	8.91	6.76	6.41	6.07	5.76	5.46	5.19
2040	23.76	15.23	9.33	8.49	7.73	7.07	6.48	5.96
2050		29.35	13.07	11.09	9.47	8.17	7.11	6.26
2060			18.96	14.34	10.98	8.62	6.98	5.84
2070			28.83	18.46	11.68	7.92	5.94	4.80
2080			35.32	23.66	10.70	6.04	4.45	3.67
2090				29.09	7.51	4.09	3.27	2.79
2100				29.61	4.07	2.93	2.47	2.14



Table 6. Changes in population (P) associated with usage rate of natural resources (NRUN) set equal to other variables after 1970.

Year	Basic Model	Population (PE 9)			
		R/R0	Usage rate (NRUN) set equal to:-		M
			R/R1970	Q	
1910	1.67	1.67	1.67	1.67	1.67
1920	1.87	1.87	1.87	1.87	1.87
1930	2.17	2.17	2.17	2.17	2.17
1940	2.56	2.56	2.56	2.56	2.56
1950	3.00	3.00	3.00	3.00	3.00
1960	3.48	3.48	3.48	3.48	3.48
1970	3.97	3.97	3.97	3.97	3.97
1980	4.43	4.44	4.43	4.43	4.43
1990	4.85	4.86	4.85	4.85	4.84
2000	5.16	5.20	5.17	5.18	5.15
2010	5.35	5.44	5.39	5.40	5.33
2020	5.40	5.55	5.48	5.50	5.38
2030	5.32	5.52	5.44	5.47	5.31
2040	5.15	5.30	5.29	5.30	5.16
2050	4.92	4.84	5.01	4.99	4.97
2060	4.71	3.97	4.64	4.52	4.79
2070	4.53	2.39	4.19	3.88	4.65
2080	4.36		3.67	3.02	4.52
2090	4.17		3.11	1.97	4.40
2100	3.97		2.55	1.35	4.27

Table 7. Changes in pollution ratio (POLR) associated with usage rate of natural resources (NRUN) set equal to other variables after 1970

Year	Basic model	Pollution ratio (POLR)			
		R/R0	Usage rate (NRUN) set equal to:-		
			R/R 1970	Q	M
1910	0.05	0.05	0.05	0.05	0.05
1920	0.09	0.09	0.09	0.09	0.09
1930	0.15	0.15	0.15	0.15	0.15
1940	0.24	0.24	0.24	0.24	0.24
1950	0.37	0.37	0.37	0.37	0.37
1960	0.57	0.57	0.57	0.57	0.57
1970	0.85	0.85	0.85	0.85	0.85
1980	1.24	1.24	1.24	1.24	1.24
1990	1.76	1.78	1.77	1.77	1.76
2000	2.44	2.51	2.46	2.47	2.42
2010	3.28	3.49	3.35	3.37	3.22
2020	4.23	4.81	4.47	4.54	4.11
2030	5.19	6.62	5.84	6.02	4.99
2040	5.96	9.22	7.49	7.88	5.70
2050	6.26	13.27	9.43	10.25	6.04
2060	5.84	20.38	11.68	13.33	5.83
2070	4.80	33.41	14.25	17.56	5.11
2080	3.67		17.16	23.67	4.18
2090	2.79		20.37	31.56	3.35
2100	2.14		23.62	30.99	2.71

Table 8. Calculated values of population, natural resources, capital investment, pollution ratio, quality of life, and material standard of living for NRUN equal to M after 1970

Year	PE 9	RE 9	CE 9	POLR	Q	M
1910	1.67	893.19	0.63	0.05	0.86	0.41
1920	1.87	883.80	0.93	0.09	1.02	0.54
1930	2.17	870.87	1.34	0.15	1.11	0.67
1940	2.56	853.38	1.88	0.24	1.13	0.79
1950	3.00	830.38	2.57	0.37	1.12	0.89
1960	3.48	801.13	3.40	0.57	1.07	0.97
1970	3.97	765.34	4.37	0.85	1.00	1.03
1980	4.43	721.32	5.41	1.24	0.93	1.06
1990	4.84	671.44	6.45	1.76	0.85	1.06
2000	5.15	619.09	7.40	2.42	0.77	1.03
2010	5.33	567.69	8.18	3.22	0.71	0.99
2020	5.38	519.85	8.73	4.11	0.66	0.93
2030	5.31	477.04	9.04	4.99	0.63	0.88
2040	5.16	439.80	9.11	5.70	0.61	0.82
2050	4.97	408.06	8.99	6.04	0.60	0.76
2060	4.79	381.46	8.70	5.83	0.59	0.70
2070	4.65	359.51	8.30	5.11	0.58	0.64
2080	4.52	341.61	7.83	4.18	0.56	0.58
2090	4.40	327.09	7.32	3.35	0.54	0.52
2100	4.27	315.29	6.80	2.71	0.52	0.47

Table 9. Optimum parameter values for substitution model, and values of fraction substituted

Parameters		Parameter values						
$\alpha = H$	0.220	0.075	0.027	0.019	0.016	0.014	0.012	0.010
$t_0 = T$	2040	2050	2060	2070	2080	2090	2100	2110
Year	Values of $f$ = fraction substituted							
1980	0.00	0.00	0.01	0.03	0.04	0.04	0.05	0.07
1990	0.00	0.00	0.02	0.05	0.05	0.06	0.07	0.08
2000	0.00	0.00	0.04	0.07	0.07	0.07	0.08	0.10
2010	0.00	0.00	0.06	0.09	0.10	0.10	0.10	0.12
2020	0.00	0.01	0.10	0.13	0.13	0.12	0.13	0.14
2030	0.01	0.05	0.17	0.18	0.17	0.16	0.16	0.17
2040	0.50	0.18	0.25	0.24	0.22	0.20	0.19	0.20
2050	0.99	0.50	0.37	0.32	0.28	0.25	0.23	0.23
2060	1.00	0.82	0.50	0.41	0.35	0.30	0.28	0.27
2070	1.00	0.95	0.63	0.50	0.42	0.36	0.33	0.31
2080	1.00	0.99	0.75	0.59	0.50	0.43	0.38	0.35
2090	1.00	1.00	0.83	0.68	0.58	0.50	0.44	0.40
2100	1.00	1.00	0.90	0.76	0.65	0.57	0.50	0.45

Table 10. Population estimates for given substitution models


Population (PE 9) for substitution models with parameters:-								
Year	H 0.220 T 2040	H 0.075 T 2050	H 0.027 T 2060	H 0.019 T 2070	H 0.016 T 2080	H 0.014 T 2090	H 0.012 T 2100	H 0.010 T 2110
1980	4.43	4.43	4.43	4.43	4.43	4.43	4.44	4.44
1990	4.85	4.85	4.85	4.85	4.85	4.85	4.85	4.85
2000	5.16	5.16	5.16	5.17	5.17	5.17	5.17	5.18
2010	5.35	5.35	5.36	5.37	5.37	5.38	5.38	5.39
2020	5.40	5.41	5.42	5.44	5.44	5.45	5.45	5.46
2030	5.32	5.33	5.36	5.38	5.38	5.39	5.39	5.40
2040	5.15	5.16	5.20	5.22	5.22	5.22	5.22	5.23
2050	5.00	4.96	5.00	5.00	4.99	4.98	4.98	4.98
2060	4.91	4.81	4.80	4.77	4.75	4.74	4.73	4.70
2070	4.87	4.73	4.66	4.60	4.57	4.56	4.53	4.48
2080	4.84	4.70	4.59	4.51	4.48	4.46	4.43	4.36
2090	3.84	4.69	4.56	4.48	4.44	4.41	4.38	4.33
2100	4.84	4.69	4.54	4.45	4.39	4.34	4.31	4.29

Table 11. Calculated values of population, natural resources, capital investment, pollution ratio, and quality of life for the optimum substitution strategy (H=0.220; T=204.0)

Year	PE 9	RE 9	CE 9	POLR	Q
1910	1.67	893.19	0.63	0.05	0.86
1920	1.87	883.80	0.93	0.09	1.02
1930	2.17	870.87	1.34	0.15	1.11
1940	2.56	853.38	1.88	0.24	1.13
1950	3.00	830.38	2.57	0.37	1.12
1960	3.48	801.13	3.40	0.57	1.07
1970	3.97	765.34	4.37	0.85	1.00
1980	4.43	723.39	5.41	1.24	0.93
1990	4.85	676.37	6.46	1.76	0.85
2000	5.16	626.05	7.43	2.44	0.78
2010	5.35	574.53	8.24	3.28	0.72
2020	5.40	523.98	8.82	4.23	0.67
2030	5.32	476.38	9.14	5.19	0.63
2040	5.15	440.23	9.19	5.97	0.61
2050	5.00	434.74	9.13	6.39	0.63
2060	4.91	434.65	9.08	6.55	0.66
2070	4.87	434.65	9.04	6.59	0.67
2080	4.84	434.65	9.01	6.55	0.67
2090	4.84	434.65	8.99	6.48	0.68
2100	4.84	434.65	8.96	6.39	0.68

Figure 1.

Main components of BASIC model



OLD

OLD PROGRAM NAME--WORLD

READY

RUN

INITIAL CONDITIONS:1900

POPULATION	1.65E9
NATURAL RESOURCES	900E9
CAPITAL INVESTMENT	0.4E9
POLLUTION	0.2E9
LAND AREA	135E6

Figure 2. Graphical plotting of levels of population, natural resources, capital investment, pollution ratio, and quality of life for basic model






Figure 3. Changes in population estimates as a  
result of changes in initial parameters of  
natural resources and capital investment



Figure 4. Changes in natural resource estimates  
as a result of changes in initial  
parameters of natural resources and  
capital investment




Figure 5. Changes in the capital investment estimates as a result of changes in initial parameter of natural resources and capital investment




Figure 6. Changes in pollution index as a result of changes in initial parameters of natural resources and capital investment




Figure 7. Changes in quality of life as a result  
of changes in initial parameters of  
natural resources and capital investment





Figure 8. Changes in P associated  
with values of NRUN



**Figure 9. Changes in POLR associated with values of NRUN**




Figure 10. Changes in population associated with  
usage normal equal to other variables








Figure 11. Changes in POLR associated with usage  
normal equal to other variables



PR9 for the year 2100



Population in  
year 2100





Figure 14. Values of H for which the population in year 2100 is a maximum at each value of T



**Figure 15. Fraction substituted by years  
for parameters of Table 2**




Figure 16. Population estimates for substitution  
models compared with basic model





Figure 17. Graphical plotting of levels of population, natural resources, capital investment, pollution ratio, and quality of life for the optimum substitution strategy ( $H = 0.220$ ;  $T = 2040$ )

RUN

INITIAL CONDITIONS:1900

POPULATION	1.65E9
NATURAL RESOURCES	900E9
CAPITAL INVESTMENT	0.4E9
POLLUTION	0.2E9
LAND AREA	135E6



READY

Appendix IListing of BASIC program

```
3 REM WORLD2 INTERACTIVE VERSION
5 REM SET INITIAL CONDITIONS
10 PRINT "INITIAL CONDITIONS:1900"
15 PRINT "POPULATION"," ", "1.65E9"
20 LET P=1.65E9
25 PRINT "NATURAL RESOURCES","900E9"
30 LET R=900E9
35 LET R0=R
40 PRINT "CAPITAL INVESTMENT","0.4E9"
45 LET C=0.4E9
50 PRINT "POLLUTION"," ", "0.2E9"
55 LET Z=0.2E9
60 PRINT "LAND AREA"," ", "135E6"
65 LET L=135E6
80 REM SET COEFFICIENTS FOR 1900 TO 1970
85 LET M=1
87 LET N5=1
90 LET L0=26.5
95 LET C4=1
100 LET C3=0.3
105 LET C2=0.2
110 LET C1=1
115 LET F5=1
120 LET F6=1
125 PRINT
130 PRINT "PE9","RE9","CE9","POLR","Q"
135 PRINT
140 REM ITERATIONS FOR 1900 TO 1970
145 FOR I=1910 TO 1970 STEP 10
150 FOR J=1 TO 10
155 GOSUB 1000
160 NEXT J
165 PRINT
170 PRINT INT((P/10^7)+0.5)/10^2,
175 PRINT INT((R/10^7)+0.5)/10^2,
180 PRINT INT((C/10^7)+0.5)/10^2,
185 PRINT INT(Z1*10^2+0.5)/10^2,
187 PRINT INT(Q*10^2+0.5)/10^2
190 NEXT I
200 REM SET COEFFICIENTS FOR 1970 ONWARDS
210 LET N5=1
300 REM ITERATIONS FOR 1970 TO 2100
305 FOR I=1980 TO 2100 STEP 10
310 FOR J=1 TO 10
315 GOSUB 1000
320 NEXT J
325 PRINT
330 PRINT INT((P/10^7)+0.5)/10^2,
335 PRINT INT((R/10^7)+0.5)/10^2,
340 PRINT INT((C/10^7)+0.5)/10^2,
345 PRINT INT(Z1*10^2+0.5)/10^2,
347 PRINT INT(Q*10^2+0.5)/10^2
360 NEXT I
995 STOP
```



```

1000 REM MAIN SUBROUTINE
1010 LET Z1=Z/3.6E9
1020 LET F4=1.035-0.01216*Z1-5.333E-4*Z1+2+7.7775E-6*Z1+3
1030 LET C7=C*0.025
1040 LET C1=C/P
1050 LET B4=1.029-9.481E-3*Z1-5.631E-4*Z1*Z1+7.778E-6*Z1*Z1*Z1
1060 LET C4=P/(L*L0)
1070 LET E3=1.048+0.028*C4-0.0764*C4*C4+0.0102*C4*C4*C4
1080 LET D4=0.964-0.132*C4+0.1036*C4+2
1090 LET D2=0.927+0.0137*Z1+2.071E-3*Z1+2
1100 LET N2=R/R0
1110 LET N1=0+N2*(-0.0666+N2*(3.2-2.133*N2))
1120 LET E=C1*(1-C2)*N1/(1-C3)
1130 LET M=E/1.0
1140 LET B2=1.2+M*(-0.225+0.025*M)
1150 LET Q4=1.05-0.0187*Z1-3.55E-4*Z1*Z1+6.39E-6*Z1*Z1*Z1
1160 LET Q2=1.98+C4*(-1.50+C4*(0.60+C4*(-0.118+9.01E-3*C4)))
1170 LET Z5=0.576+0.1704*Z1+2.600E-3*Z1*Z1
1180 LET Z3=Z/Z5
1190 LET N4=0.114+M*(0.899-0.052*M)
1200 LET Q1=0.179+0.924*M-0.075*M*M
1210 LET Z4=0.062+0.051*C1+0.986*C1*C1-0.136*C1*C1*C1
1220 LET Z2=P*1*Z4
1230 LET C8=0.082+1.04*M-0.091*M*M
1240 LET C6=P*C8*0.05
1250 LET C5=C1*C2/C3
1260 LET F2=0.520+0.504*C5-0.0381*C5*C5
1270 LET F3=2.40-2.158*C4+0.955*C4+2-0.1955*C4+3+0.01465*C4+4
1280 LET F1=F2*F3*F4*F5/F6
1290 LET B1=0.02+1.09*F1-0.15*F1*F1
1300 LET D3=3.8125-3.998*F1+1.19*F1*F1
1310 LET D1=0.5178+2.521*0.2186147+M
1320 LET D=P*0.028*D1*D2*D3*D4
1330 LET N3=P*N5*N4
1340 LET B=P*0.04*B1*B2*B3*B4
1350 LET Q3=-8.57E-3+F1*(1.137-0.1143*F1)
1360 LET Q=1*C1*Q2*Q3*Q4
1370 LET A1=1.001-0.936*F1+0.243*F1*F1
1380 LET A2=0.697+0.0314*(Q1/Q3)+0.3143*(Q1/Q3)+2
1390 LET C2=C2+(1/15)*(A1*A2-C2)
1400 LET C=C+(C6-C7)
1410 LET R=R-N3
1420 LET P=P+(B-D)
1430 LET Z=Z+(Z2-Z3)
1490 RETURN
1600 END

```

READY